
概 要 継続とは、ある時点における残りの計算のことである。プログラム内で継続を
扱う方法の一つに、プログラム全体を継続渡し形式 (Continuation-Passing Style; CPS)
に変換するというものがある。本研究では、単純型付き λ計算におけるDanvy/Filinski
による one-passの CPS変換の正当性を定理証明支援系 Agdaによって定式化し、その
正当性を証明する。さらに、この証明が let多相を含む体系でも成り立つよう拡張を行
う。このとき、変数名を明示的に扱うのではなく、変数の束縛をメタ言語で行う PHOAS
(Parameterized Higher-Order Abstract Syntax)を用いて証明することにより、手で書
くのと同程度の複雑さで定式化することができた。

1 はじめに
継続とは、ある時点における残りの計算のことである。例えば、1 + (2 ∗ 3)という計算で、2 ∗ 3

を実行しているときの継続は、「2 ∗ 3の計算結果を受け取ったら、そこに 1を加える」ということに
なる。プログラムの中で継続を扱う方法に、プログラム全体を継続渡し形式 (Continuation-Passing

Style; CPS)に変換するというものがある。CPS変換をするとプログラム中の継続を管理できるた
め、コンパイラの中間言語に使われる [1] など数々の応用がある。
本稿は、selective CPS変換 [13] のAgda [17] による定式化を試みる研究の一環で、単純型付き λ

計算に let多相と限定継続のオペレータ shift/reset [8] を加えた体系 [2,8] での selective CPS 変
換の証明 [3] の定式化を目指している。shift/reset は限定継続を取り扱うためのオペレータであ
り、shiftが「その時点の継続を切り取ってくる命令」で、resetが「切り取られる継続の範囲を限
定する命令」である。shift/resetには、非決定性プログラミング [8]や let insertion を用いた部
分評価 [11] などいろいろな応用例がある。shift/resetを使ったプログラムは CPS 変換を行えば
shift/resetをサポートしない言語でも実行可能だが、プログラム全体を CPS変換すると効率が落
ちる。selective CPS変換では、shift/resetに必要な部分だけ CPS 変換することで shift/reset

に関係ない部分の効率を落とすことなく実行できるようになる [3]。
shift/resetを導入するためには、現在のコンテキストの型をあらわす answer typeについても

各種の証明をしなければならない。さらに、selective CPS変換は与えられた項の answer typeが変
化するかどうかによって数多くの場合分けをする必要があり、手で証明するのは次第に困難になっ
てくる。そのような証明を Agda で行えるようにするべく、本稿ではその基礎となる let 多相の入っ
た単純型付き λ計算を対象にして、その CPS 変換の正当性の証明を Agda で行う。
これまで、各種のプログラミング言語理論の定式化を行うためには、変数束縛の問題を細部にわ

たって正確に管理しなければならなかった。しかし、PHOAS (Parameterized Higher-Order Abstract

Syntax) [5] を使うと、変数をある程度、明示的に扱いつつも変数束縛をメタ言語で取り扱うこと
ができるようになるため、変数束縛の管理にそれほど大きな労力を割くことなく定式化できる。特
に、多相の入った体系でも PHOASの利点を生かして証明できるという見通し [5]が得られており、
selective CPS変換の定式化への基盤として利用するにはPHOASが優れていると考えられる。本稿
では、実際に CPS変換の正当性の証明を行ってみた様子を報告し、let多相の入っていない体系で
は、十分に簡潔で見通しの良い形で証明できることを示す。

一方、PHOASは HOAS (Higher-Order Abstract Syntax)の一つなので、Agdaで λ式の中を分
解できないなどの困難な点もある。実際、let多相の入った体系の CPS変換の証明を行う際には、
自明に成り立つと思われる性質を PHOASを使った定式化では示すことができず、仮定せざるを得
なかった。その様子も本稿で報告する。
PHOAS はすでに Chlipala 自身によってコンパイラの正当性の表示的意味論に基づいた証明に

使用されている。その中で System F の CPS 変換が扱われており [5]、その意味では本稿の結果は
予想されるものである。しかし、Danvy/Filinski の one-passの CPS変換の正当性の証明 [9] をほ
ぼそのまま Agda で定式化できており、PHOAS の適用範囲をみるケーススタディとしては一定の
価値があると思われる。また、let 多相の定式化は System F とは違った形の定式化になっており、
そこには若干の新規性があると考えている。let多相の定式化には、仮定が残っているなど HOAS

特有の問題点は存在するが、それを除くと十分に簡潔な証明になっており、今後、限定継続命令へ
の拡張などに道を開くものとなっている。
以下、2節で単純型付き λ計算を、3節でDanvy/Filinskiの one-passのCPS変換をAgdaを用い

て定式化した上で、4節で CPS変換の正当性を証明する。加えて、5節で単純型付き λ計算に let

多相を加えて定式化し、PHOAS特有の問題はあるものの、仮定を立てることで証明を試みる。ま
た、6章で関連研究を、7節でまとめと今後の課題を述べる。
本稿で示した証明の Agda のコードは http://pllab.is.ocha.ac.jp/~asai/jpapers/ppl/18/

に置いている。

2 単純型付きλ計算の定式化
この節では、対象言語を導入し、その PHOAS を使った Agda による定式化について述べる。

2.1 PHOAS

Agda で定式化するにあたって、PHOAS (Parameterized Higher-Order Abstract Syntax)を用い
る。PHOASとは、変数束縛のある言語の抽象構文木を表現するための方法の一つである。
対象言語の変数束縛は通常、名前を使って行われる。このような一階の抽象構文は First-Order

Abstract Syntax (FOAS) と呼ばれる。FOAS を使うと、簡単に変数を扱える反面、変数のα同
値性を自分で定義する必要がある。これは、ほとんどの場合とても煩雑で、その解決を目指して
PoplMark Challenge [4]が提示されるなど多くの解が求められてきた。PoplMark Challenge

は、多相の λ計算に部分型付けを組み合わせた体系 F<:において、変数束縛や複雑な帰納法の問題
などを取り扱うことを目的として設けられている。
Higher-Order Abstract Syntax (HOAS) は、変数束縛の問題を回避するひとつの方法で、名前解

決を対象言語ではなくメタ言語 (本稿では Agda) で行う。メタ言語の変数束縛をそのまま対象言語
の定義に使用するため、定義が簡潔になり、代入などの操作も行いやすくなる。一方で、メタ言語
の関数をそのまま使用するため、関数の中身に従った場合分けができなくなるなどの問題点もある。
PHOAS はその問題点を、変数をパラメータ化することで一部、緩和するものである。

2.2 単純型付き λ計算
本稿で対象とする言語は純粋な単純型付き λ計算に定数 (自然数)を加えたものである。その型と

構文を図 1に示す。型は自然数型か関数型のどちらかである。これは Agda では以下のように定義
できる。

data typ : Set where
Nat : typ �� 自然数型
_⇒_ : typ → typ → typ �� 関数型

型 τ ::= Nat | τ → τ

値 v ::= c | x | λx. e
項 e ::= v | e1 e2

図 1. 単純型付き λ計算

値は変数・自然数・λ抽象の 3つからなり、項は値か関数適用である。本稿では、このように定
義される構文のうち型が付くもののみを扱う。Agda では、依存型を使って「型 τ を持つ項」など
を表現できる。これを使って、次のように値と項を Agda で定義する。値と項は相互再帰的に定義
されるため、mutualというキーワードを使って定式化する。

mutual
data value[_]_ (var : typ → Set) : typ → Set where

Var : {τ1 : typ} → var τ1 → value[ var ] τ1 �� 変数
Num : (n : N) → value[ var ] Nat �� 自然数
Fun : {τ1 τ2 : typ} → �� ラムダ抽象

(e1 : var τ2 → term[ var ] τ1) → value[ var ] (τ2 ⇒ τ1)

data term[_]_ (var : typ → Set) : typ → Set where
Val : {τ1 : typ} → value[ var ] τ1 → term[ var ] τ1 �� 値
App : {τ1 τ2 : typ} → �� 関数適用

(e1 : term[ var ] (τ2 ⇒ τ1)) → (e2 : term[ var ] τ2) → term[ var ] τ1

value[var ]τ と term[var ]τ は、それぞれ τ 型の値、項を示す。ここで var は、変数を表すパラ
メータで typ を引数に取る。ここは PHOAS 特有の部分である。var は型環境のことだと考えても
良い。Var x は、x が var τ1 型のとき（x が型環境で τ1 型を持つとき）τ1 型の変数を表す。Num

は、自然数 n を受け取ると無条件で Nat 型の定数となる。Fun は、「τ2 型の変数を受け取ったら τ1

型の項を返す関数」を受け取ったら、(τ2 ⇒ τ1) 型の λ抽象となる。ここで Fun は、Agda 上の関
数を引数に取ることに注意しよう。対象言語の変数の束縛をメタ言語 (Agda) の束縛で表現してお
り、ここが HOAS になっている部分である。例えば、λx. λy. x は Fun (λx → Val (Fun (λy → Var

x))) と表される。通常の HOAS では引数も項となるが、ここでは引数の型は変数 var τ1 となって
いる。値と項全体が var でパラメータ化されているため、変数も独立した構文（構成子）として扱
うことができるようになるとともに、関数の引数部分（negative な部分）に値や項自身が現れるこ
とを避け、Agda でも構文を定義できるようになっている。これが PHOAS である。最後に、App

は、(τ2 ⇒ τ1) 型の項と τ2 型の項を受け取ると、全体として τ1 型の関数適用となる。
上記の値、項の定義は単純型付き λ計算の型規則をそのまま埋め込んだものになっている。この

ように構文を定義すると、型の付く構文のみを議論の対象とすることができる。

2.3 代入規則
のちに正当性の証明に β簡約を利用するため、2.2節の体系での代入規則を relationで定義したも

のを図 2に示す。(λy. e)[v] 7→ e′と書くと、「項 eの中に出てくる yを値 vで置き換えると項 e′とな
る」と読み、Agdaでは Subst e v e′と書く。項 eの中に出てくる yと書くと、eの中に yが自由に現
れている必要があるが、PHOAS を使っていると自由に現れる変数を表現できない。そこで (λy. e)

のように、代入する式が入る場所 yを λ式で抽象化した式を使っている。(λy. e)[v] 7→ e′の各場合
については、以下に示すAgdaのコードを参照しながら説明する。代入先の valueや termは、2.2節
では相互再帰によって定義されているため、代入規則もそれに倣った構造となる。
SubstValは、「代入する場所が τ 型であるような τ1型の値」と τ 型の値を受け取ったら、値を代

入した結果として τ1型の値を返す規則である。sVar=は、変数が代入先の変数だった場合で、vを

(λy. y)[v] 7→ v (λy. x)[v] 7→ x (λy. c)[v] 7→ c
∀x((λy. (e y)x)[v] 7→ e′ x)

(λy. λx. e y)[v] 7→ λx. e′

(λy. e1 y)[v] 7→ e′1 (λy. e2 y)[v] 7→ e′2
(λy. (e1 y) (e2 y))[v] 7→ e′1 e

′
2

図 2. 代入規則

mutual
data SubstVal {var : typ → Set} : {τ τ1 : typ} →

(var τ → value[ var ] τ1) → value[ var ] τ → value[ var ] τ1 → Set where
sVar= : {τ : typ} → {v : value[ var ] τ} → �� 変数に代入する

SubstVal (λ x → Var x) v v
sVar̸= : {τ τ1 : typ} → {x : var τ1} → {v : value[ var ] τ} → �� 変数に代入しない

SubstVal (λ y → Var x) v (Var x)
sNum : {τ : typ} → {n : N} → {v : value[ var ] τ} → �� 自然数には代入しない

SubstVal (λ y → Num n) v (Num n)
sFun : {τ τ1 τ2 : typ} → �� ラムダ抽象に代入する

{e1 : var τ → var τ2 → term[ var ] τ1} → {v : value[ var ] τ} →
{e1

′ : var τ2 → term[ var ] τ1} →
((x : var τ2) → Subst (λ y → (e1 y) x) v (e1

′ x)) →
SubstVal (λ y → Fun (e1 y)) v (Fun e1

′)

data Subst {var : typ → Set} : {τ τ1 : typ} →
(var τ → term[ var ] τ1) → value[ var ] τ → term[ var ] τ1 → Set where

sVal : {τ τ1 : typ} → �� 値に代入する
{v1 : var τ → value[ var ] τ1} → {v : value[ var ] τ} → {v1

′ : value[ var ] τ1} →
SubstVal v1 v v1

′ →
Subst (λ y → Val (v1 y)) v (Val v1

′)
sApp : {τ τ1 τ2 : typ} → �� 関数適用に代入する

{e1 : var τ → term[ var ] (τ2 ⇒ τ1)} → {e2 : var τ → term[ var ] τ2} →
{v : value[ var ] τ} → {e1

′ : term[ var ] (τ2 ⇒ τ1)} → {e2
′ : term[ var ] τ2} →

Subst e1 v e1
′ → Subst e2 v e2

′ →
Subst (λ y → App (e1 y) (e2 y)) v (App e1

′ e2
′)

xに代入して値 vを返す。sVar ̸=は、変数が代入先の変数とは異なった場合で、値 vは代入されず変
数 xを返す。sNumは、自然数には代入先の変数は現れないので元の自然数がそのまま返ることを
示す。sFunは、少し面倒な形をしているが、以下のように理解できる。まず、λ抽象の本体部分 e y

はもともと PHOAS のため引数 xを受け取るような関数で表現されている。そこに、さらに代入す
る場所を示すために yが導入されるため、eは結果として yと xを受け取るような関数となる。そ
の y部分に vを代入するためには、任意の xについて (e y)x の y部分に vを代入した結果を得れば
良い。ここで、任意の xを選ぶ部分はメタ言語で行われていることに注意しよう。ここでもHOAS

の考え方が使われている。
次にSubstを見る。sValでは、値 v1の中に現れる yに値 vを代入したら値 v′1になることをSubstVal

を用いて定義している。sAppでは、項 e1, e2の中に現れる yに値 vを代入したらそれぞれ項 e′1, e
′
2

になるとき、関数適用 e1 e2は e′1 e
′
2になることを示している。

上記の代入規則は型付きで定義されているため、代入規則を Agda で定義できた時点で次の代入
補題 [18] が示されたことになる。

補題 1 (代入の下での型の保存) Γ, y : τ ′ ⊢ e : τ かつ Γ ⊢ v : τ ′ならば Γ ⊢ (λy. e)[v] : τ である。

本研究では代入規則を relation として定義しているが、これ以外の方法としてはAgdaの関数を
用いて定義する [5,6]ことも考えられる。関数による定義は、直接、関数の結果として代入結果を得
られるが、その際 var を具体化する必要がある。すると、small-step の意味論では、簡約をするた

びに var の具体化と一般化を繰り返さなければならず、煩雑になる。そのように定式化しても証明
は可能かもしれないが、本論文では relation を使って代入規則を定義する方法をとった。Chlipala

自身も System F の定式化の際には型変数への代入を relation で定義している [5]。

2.4 簡約規則
代入規則に続いて β簡約規則を定める。簡約は call-by-valueで行うものとする。call-by-valueと

は、関数適用の際に引数を先に評価してから簡約をすることである。
まず β簡約規則のための評価規則を図 3に示すフレームと評価文脈を用いて定義する。[ ]はこれ

から評価を行う場所を意味する。フレームと、フレームの [ ]に項を入れる frame-plugはAgdaで次
のように書ける。

data frame[_,_] (var : typ → Set) : typ → typ → Set where
App1 : {τ1 τ2 : typ} → (e2 : term[ var ] τ2) → frame[ var , τ2 ⇒ τ1 ] τ1
App2 : {τ1 τ2 : typ} → (v1 : value[ var ] (τ2 ⇒ τ1)) → frame[ var , τ2 ] τ1

frame-plug : {var : typ → Set} → {τ1 τ2 : typ} →
frame[ var , τ2 ] τ1 → term[ var ] τ2 → term[ var ] τ1

frame-plug (App1 e2) e1 = App e1 e2
frame-plug (App2 v1) e2 = App (Val v1) e2

frame[var, τ2] τ1は「全体として τ1型になるような項で、中の [ ]が τ2型をもつフレーム」を意味
する。App1は関数部分を実行中であることを示すフレームである。App e1 e2という関数適用があ
り e1が項のときは、e1が先に評価される。App2は引数部分を実行中であることを示すフレームで
ある。App v1 e2という関数適用で v1が値、e2が項のとき、v1は評価済みなので e2が評価される。
次に、図 4のような β簡約規則を定義する。簡約規則には、1ステップ簡約する Reduceと、0ス

テップ以上簡約する Reduce*の 2つの規則がある。
data Reduce {var : typ → Set} : {τ1 : typ} → term[ var ] τ1 → term[ var ] τ1 → Set where

RBeta : {τ1 τ2 : typ} → �� β 簡約
{e : (x : var τ2) → term[ var ] τ1} → {v2 : value[ var ] τ2} → {e′ : term[ var ] τ1} →
Subst e v2 e′ → Reduce (App (Val (Fun e)) (Val v2)) e

′

RFrame : {τ1 τ2 : typ} → �� CBVによる部分式の評価
{e e′ : term[ var ] τ1} → (f : frame[ var , τ1 ] τ2) →
Reduce e e′ → Reduce (frame-plug f e) (frame-plug f e′)

data Reduce* {var : typ → Set} : {τ1 : typ} → term[ var ] τ1 → term[ var ] τ1 → Set where
R*Id : {τ1 : typ} → (e : term[ var ] τ1) → Reduce* e e �� 0ステップの簡約
R*Trans : {τ1 : typ} → (e1 e2 e3 : term[ var ] τ1) → �� 1ステップ以上の簡約

Reduce e1 e2 → Reduce* e2 e3 → Reduce* e1 e3

RBetaは、関数適用を実行する。ここで、項 eは xを受け取って τ1型の項を返すような関数抽象で
ある。eの中に出てくるxに値 v2を代入した結果が e′になるとき、(App (Val (Fun e))(Val v2))を実行
すると e′となる。RFrameはフレームを用いた評価規則に基づいて項を評価する。ここで frame-plug

f eは項 eにフレーム f をつけることを意味していたことに注意すると、項 eを簡約した結果が e′

になるとき、f [e]は f [e′]に簡約される。
次に Reduce*を見てみる。Reduce*は Reduceの反射推移閉包である。R*Idは、受け取った項 eは

0ステップで自分自身となることを意味する。R*Transは、1ステップ以上、簡約を進める規則であ
る。項 e1を 1ステップ簡約した結果が項 e2となり、項 e2を 0ステップ以上簡約した結果が e3にな
るとき、e1を 1ステップ以上簡約した結果が e3になることを意味する。
ここでも、簡約規則がすべて型付きで定義されているため、簡約規則を Agdaで書けたことで、

項を簡約した結果は型が保存されていること、つまり subject reduction [16] が証明されたことに
なる。

フレーム F = [ ] e | v [ ]
評価文脈 E = [ ] | F [E]

図 3. 評価文脈

(λx. e)[v] 7→ e′

(λx. e) v → e′
e → e′

F [e] → F [e′]

図 4. 簡約規則

3 CPS変換の定式化
本稿では、Danvy/Filinski による one-passの CPS 変換 [9] を扱う。ここでは call-by-value かつ

left-to-right の CPS 変換を扱う。型の変換を図 5に、項の変換を図 6に示す。本稿では、answer

type は Nat に固定する。

Nat∗ = Nat

(τ1 ⇒ τ2)
∗ = τ1

∗ ⇒ (τ2
∗ ⇒ Nat) ⇒ Nat

図 5. 型の CPS変換

[[c]]v = c

[[x]]v = x

[[λx. e]]v = λx. λk. [[e]]′@ k

[[v]] = λκ. κ@ [[v]]v

[[e1@ e2]] = λκ. [[e1]] @ (λm. [[e2]] @ (λn. (m@n)@ (λa. κ@ a)))

[[v]]′ = λk. k@ [[v]]v

[[e1@ e2]]
′ = λk. [[e1]] @ (λm. [[e2]] @ (λn. (m@n)@ k))

図 6. 単純型付き λ計算の CPS変換

CPS 変換は、値用の [[v]]v と項用の [[e]] と [[e]]′ の３種類からなる。CPS 変換の結果は、アンダー
ラインのついた式とオーバーラインのついた式を使った 2 レベルの λ計算で書かれている。アン
ダーラインのついた式は dynamic な式と呼ばれ、出力の構文を表す。Agda では Fun や App など
の構成子で表現される。一方、オーバーラインのついた式は static な式と呼ばれ、CPS 変換時に
これらの命令は実行されることを示す。これらは Agda の（実行可能な）関数や関数呼び出しとな
る。static な命令を CPS 変換時に実行してしまうため、いわゆる administrative β-redex を生じ
ることなく、コンパクトな CPS 変換結果を生成することができる [9]。
[[e]]κは、項 eを受け取ると、それを static な初期継続 κのもとで CPS変換した結果を返すこと

を意味する。このとき κはメタ言語で書かれる関数であり、変換時に実行される。一方、[[e]]′ kは、
項 eを受け取ると、それを dynamic な初期継続 kのもとでCPS変換した結果を返すことを意味す
る。このように２種類の変換を用意しているのは、CPS 変換時に継続がわかっている場合とわかっ
ていない場合があるためである。関数適用の規則では継続は具体的な形で与えられているが、λ抽
象の規則では継続の形は不明である。このように継続の形がわかっているかどうかで２種類の変換
を用意することで、administrative η-redex を生じることなく CPS 変換結果を得ることができるよ
うになる [9]。
上の規則を Agdaで実装すると以下のようになる。規則をほぼそのまま Agdaに直せていることが

わかる。cpsEta と cpsEta′ の継続は、前者は static なためAgda の関数型を持ち、後者は dynamic

なため（関数型を持つ）値となっている。PHOAS 特有のこととして、λ抽象の場合は少し説明が
必要である。PHOAS を使っているため、Fun の引数である e1 は static な関数である。また、CPS

�� 型レベルの CPS変換
cpsT : typ → typ
cpsT Nat = Nat
cpsT (τ2 ⇒ τ1) = cpsT τ2 ⇒ (cpsT τ1 ⇒ Nat) ⇒ Nat

mutual
�� 値に対する CPS変換
cpsEtaV : {var : typ → Set} → {τ : typ} → value[ var ◦ cpsT ] τ → value[ var ] cpsT τ
cpsEtaV (Var x) = Var x
cpsEtaV (Num n) = Num n
cpsEtaV (Fun e1) = Fun (λ x → Val (Fun (λ k → cpsEta′ (e1 x) (Var k))))

�� 項に対する CPS変換（static な継続）
cpsEta : {var : typ → Set} → {τ : typ} →

term[ var ◦ cpsT ] τ → (value[ var ] (cpsT τ) → term[ var ] Nat) → term[ var ] Nat
cpsEta (Val v) κ = κ (cpsEtaV v)
cpsEta (App e1 e2) κ =

cpsEta e1 (λ v1 → cpsEta e2 (λ v2 →
App (App (Val v1) (Val v2)) (Val (Fun (λ v → κ (Var v))))))

�� 項に対する CPS変換の補助規則（dynamic な継続）
cpsEta′ : {var : typ → Set} → {τ : typ} →

term[ var ◦ cpsT ] τ → value[ var ] (cpsT τ ⇒ Nat) → term[ var ] Nat
cpsEta′ (Val v) k = App (Val k) (Val (cpsEtaV v))
cpsEta′ (App e1 e2) k =

cpsEta e1 (λ v1 → cpsEta e2 (λ v2 → App (App (Val v1) (Val v2)) (Val k)))

変換結果の Fun も引数として static な関数を受け取る必要がある。CPS 変換結果の関数の引数 x

を e1 に渡すことで e1 の本体を得て、それを CPS 変換にかけている。この変換を通じて、変数の
束縛がメタ言語の変数の束縛を通じてうまく表現されていることがわかる。
上記の Agda による CPS 変換は、変換される項の型情報も含めて定義されているため、実装が

できた時点で、次の定理が示されたことになる。

定理 1 (CPS変換の型保存) Γ ⊢ v : τ ならば Γ∗ ⊢ [[v]]v : τ∗ が成り立つ。Γ ⊢ e : τ ならば
Γ∗ ⊢ [[e]] : (τ∗ → Nat) → Nat かつ Γ∗ ⊢ [[e]]′ : (τ∗ ⇒ Nat) → Nat が成り立つ。

CPS変換の例を考える。[[3]]は λκ. κ@3になるが、これは cpsEta (Val (Num3))を実行すると
λκ → κ (Num3)になるのと等しい。また、[[(λx. x)@ 3]]は

λκ. ((λx. λk. k@x)@ 3)@ (λv. κ@ v)

になる。これは、cpsEta (App (Val (Fun (λx→ (Val (Var x))))) (Val (Num3)))が

λκ → App (App (Val (Fun (λx → Val (Fun (λk → App (Val (Var k)) (Val (Var x)))))))

(Val (Num3))) (Val (Fun (λv→ κ (Var v))))

となるのと同じである。

4 単純型付きλ計算のCPS変換の正当性の証明
この節では、3 節で示した CPS 変換が正しいこと、つまり CPS 変換をしても、変換前の項と同

等の簡約を行うことを示す。最終的に示される定理は以下のようになる。

定理 2 (CPS 変換の正当性) 任意の項 e, e′について e → e′ が成り立つなら、任意の schematic な
継続 κ について [[e]] @κ →∗ [[e′]] @κ が成り立つ。

ここで、e → e′ は入力言語での β 簡約、[[e]] @κ →∗ [[e′]] @κ は出力言語（アンダーラインのつい
た項）での β簡約 である。継続が schematic であることの定義は、4.1節で述べる。上の定理は、
Agda では以下のように書ける。

correctEta : {var : typ → Set} → {τ : typ} → {e e′ : term[ var ◦ cpsT ] τ} →
(κ : value[ var ] cpsT τ → term[ var ] Nat) →
Reduce e e′ → schematic κ → Reduce* (cpsEta e κ) (cpsEta e′ κ)

4.1 schematic

定理 2は任意の継続 κ で成り立つわけではない。例えば、e が (λx. x)@ 3、e′ が 3 だったとし
よう。static な継続 κ のもとでのそれぞれの CPS 変換は、3節で述べたように

[[e]]κ = [[(λx. x)@ 3]]κ = ((λx. λk. k@x)@ 3)@ (λv. κ@ v)

[[e′]]κ = [[3]]κ = κ@3

となる。ここで [[e]]κ →∗ [[e′]]κ となるかを調べてみると、[[e]]κ に対して３回、dynamic な β簡約 を
すると [[e′]]κ が得られるので、任意の κに対して定理が成り立っているように見える。しかし、こ
こで注意しなくてはいけないのは上の CPS 変換結果に現れる κ@ v と κ@3 は static な関数適用
で、CPS 変換時に簡約されるということである。ここで、例えば κが「渡された引数が変数だった
ら 1 を返し、それ以外だったら 2 を返す」ような継続 κ0だったとすると、上の結果は以下のよう
になってしまう。

[[e]]κ0 = [[(λx. x)@ 3]]κ0 = ((λx. λk. k@x)@ 3)@ (λv. 1)

[[e′]]κ0 = [[3]]κ0 = 2

これでは明らかに [[e]]κ0 を簡約しても [[e′]]κ0 には至らない。
上の例で定理が成り立たなかったのは κ が引数の変数を単なるデータとしてとらえており、値

の代入などが起こりうる構文木としての性質を考慮していなかったためである。そのような「引数
の syntactic な構造に独立な継続（構造を変更しないような継続）」のことを schematic な継続と呼
ぶ [9]。この性質は以下のように書ける。

(λy. κ@ y)[v] 7→ κ@ v

つまり、κ に y を渡した結果の項の y の部分に v を入れたものは、最初から κ に v を渡したもの
と等しくなるということである。
これを Agda で表現すると以下のようになる。

schematic : {var : typ → Set} → {τ : typ} →
(κ : value[ var ] cpsT τ → term[ var ] Nat) → Set

schematic {var} {τ} κ = (v : value[ var ] cpsT τ) → Subst (λ y → κ (Var y)) v (κ v)

4.2 正当性の証明
定理 2の証明の概略を示す。定理 2は、Reduce e e′の構造によって場合分けをして証明する。最初

の場合分けは帰納法の base caseに相当し、eが (λx. e1)@ vの形のとき β簡約をする。残りのケー
スはフレームの中が簡約される場合で、フレームの形によってさらに場合分けし、どちらも再帰す
る形で証明する。
各場合の証明は、手で書いた証明に近い形で書けるよう、equational reasoning に似せて証明を

書いた。結論の Reduce* 関係を示すときは、全体を begin と ■ で囲み、中に簡約したい式を順に
書き、各式の間に式変形の根拠を書く。式変形の根拠は 3 種類あり、≡⟨ ⟩ は前後の式が等しいこと
を、→⟨ ⟩ は 1ステップで簡約できることを、→∗⟨ ⟩ は 0ステップ以上で簡約できることを意味す
る。いずれも ⟨ と ⟩ の間に証明項を書く。

base caseの証明は以下のようになる。
[[(λx. e1)@ v]]κ = ((λx. λk′. [[e x]]′ k′)@ [[v2]]v)@ (λv. κ@ v)

→ (λk′. [[e′]]′ k′)@ (λv. κ@ v)

→ [[e′]]′ (λv. κ@ v)

→∗ [[e′]]κ

上の数式をそのままAgdaで定式化することができている。
correctEta {e′ = e′} κ (RBeta {e = e} {v2} sub) sch-k = �� base case

begin
cpsEta (App (Val (Fun e)) (Val v2)) κ

≡ ⟨ re� ⟩
App

(App (Val (Fun (λ x → Val (Fun (λ k′ → cpsEta′ (e x) (Var k′)))))) (Val (cpsEtaV v2)))
(Val (Fun (λ v → κ (Var v))))

−→⟨ RFrame (App1 (Val (Fun (λ v → κ (Var v)))))
(RBeta (sVal (sFun (λ k′ → eSubst′ k′ sub)))) ⟩ �� 1 ステップ簡約
frame-plug (App1 (Val (Fun (λ v → κ (Var v))))) (Val (Fun (λ k′ → cpsEta′ e′ (Var k′))))

≡ ⟨ re� ⟩
App (Val (Fun (λ k′ → cpsEta′ e′ (Var k′)))) (Val (Fun (λ v → κ (Var v))))

−→⟨ RBeta (kSubst′ e′) ⟩ �� 1 ステップ簡約
cpsEta′ e′ (Fun (λ v → κ (Var v)))

−→*⟨ cpsEtaEta′ e′ κ sch-k ⟩ �� 0 または 1 ステップ簡約
cpsEta e′ κ

■

base caseに注目すると、→⟨ ⟩が 2回使われている。このうち最初に出てくる→⟨ ⟩では、引数 x

として (Val (cpsEtaV v2))を受け取ってきて代入し、(e x)を最終的には e′に簡約している。しかし
xは CPS変換規則の cpsEta′の引数の中に現れるため 2.3節で定めた代入規則では代入ができず簡
約に進めない。これを解決するため eSubst′という補題を使っている。また、次に出てくる→⟨ ⟩で
は、引数 k′として (Val (Fun (λ v → k (Var v)))) を受け取り代入しようするが、前の例と同様 k′が
cpsEta′ の引数の中に現れるため代入ができない。そのため、kSubst′という補題を示して利用して
いる。以下に示した eSubst′と kSubst′は、いずれも CPS変換規則の cpsEta′にかかわる補題だが、
3節で示したようにCPS変換規則は相互再帰的に 3種類あるので、補題もそれにしたがってそれぞ
れ 3つずつ定める必要がある。

補題 2 (CPS変換と代入演算の可換性 1) 任意の項 e, e′、値 v、dynamicな継続 kについて
(λy. e y)[v] 7→ e′が成り立つとき、(λy. [[e y]]′ k)[[[v]]v] 7→ [[e′]]′ k が成り立つ。

eSubst′ : {var : typ → Set} → {τ1 τ2 : typ} →
{e : var (cpsT τ2) → term[ var ◦ cpsT ] τ1} →
{e′ : term[ var ◦ cpsT ] τ1} → {v : value[ var ◦ cpsT ] τ2} →
(k : var (cpsT τ1 ⇒ Nat)) →
Subst e v e′ →
Subst (λ y → cpsEta′ {var} (e y) (Var k)) (cpsEtaV v) (cpsEta′ e′ (Var k))

補題 3 (CPS変換と代入演算の可換性 2) 任意の項 e、値 v、dynamicな継続 kについて
(λk. [[e y]]′ k)[v] 7→ [[e y]]′ v が成り立つ。

kSubst′ : {var : typ → Set} → {τ : typ} →
(e : term[ var ◦ cpsT ] τ) → {v : value[ var ] (cpsT τ ⇒ Nat)} →
Subst (λ k1 → cpsEta′ e (Var k1)) v (cpsEta′ e v)

base caseの最後に現れる→∗⟨ ⟩では、e′の構造によって簡約のステップ数が異なるため以下に
示す cpsEtaEta′という補題で別途証明をしている。e′が valueのときは 1ステップの簡約が必要で、

そうでないとき簡約は必要ない。このことから、一番初めに与えられた式 (cpsEta e k) を 2ステッ
プもしくは 3ステップ簡約することで (cpsEta e′ k)に到達することがわかる。

補題 4 (0または 1ステップの簡約) 任意の項e、schematicな継続κについて [[e]]′ (λv. κ v) →∗ [[e]]κ

が成り立つ。

cpsEtaEta′ : {var : typ → Set} → {τ : typ} → �� 0 または 1 ステップの簡約
(e : term[ var ◦ cpsT ] τ) →
(κ : value[ var ] cpsT τ → term[ var ] Nat) → schematic κ →
Reduce* (cpsEta′ e (Fun (λ v → κ (Var v)))) (cpsEta e κ)

次に、フレームの中が簡約される場合は以下のようになり、これも Agdaでそのままコードにす
ることができた。

• inductive case 1

[[[ e ] e2]]κ = [[e]] (λv1. [[e2]] (λv2. (v1@ v2)@ (λv. κ@ v)))

→ [[e′]] (λv1. [[e2]] (λv2. (v1@ v2)@ (λv. κ@ v))) (IH)

= [[[ e′ ] e2]]κ

• inductive case 2

[[v1 [ e ]]]κ = [[e]] (λv2. ([[v1]]v @ v2)@ (λv. κ@ v))

→ [[e′]] (λv2. ([[v1]]v @ v2)@ (λv. κ@ v)) (IH)

= [[v1 [ e
′ ]]]κ

5 let多相による拡張
この節では、今までの節で定式化した単純型付き λ計算の定式化を、let多相によって拡張した定

式化について述べる。

5.1 多相型
let多相の定式化をするためにはまず、型を多相に変更する必要がある。多相の型は、計算が行わ

れるまでは「任意の型 α」を持ち、具体的な項において型付けを行う際に単相の型へ具体化される。
2節で定めた型 typに型変数を加えたものを単相の型とし、さらに型スキームを定義する。型ス

キームは、単相の型か多相の型からなる。Agdaによる型と型スキームの定義を以下に示す。
data typ[_] (tvar : Set) : Set where

TVar : tvar → typ[ tvar ] �� 型変数
Nat : typ[ tvar ] �� 自然数型
_⇒_ : typ[ tvar ] → typ[ tvar ] → typ[ tvar ] �� 関数型

data ts[_] (tvar : Set) : Set where
Typ : typ[ tvar ] → ts[ tvar ] �� 単相型
Forall : (ts : tvar → ts[ tvar ]) → ts[ tvar ] �� 多相型

2節の型定義と異なり、typは tvar でパラメータ化されている。この tvar は、型変数を表してお
り、型変数 TVarは tvar を受け取るように定義される。型スキーム tsも、tvar でパラメータ化さ
れている。多相の型は Forallで定義されていて、型変数を受け取ったら型スキームを返す、という
Agda上の関数で定義され、メタ言語の束縛で型変数の束縛を表現している。

型 τ ::= α | Nat | τ → τ

型スキーム σ ::= τ | ∀α.σ
値 v ::= c | x | λx. e
項 e ::= v | e1 e2 | let x = v in e x

図 7. let多相を加えた単純型付き λ計算

Γ ⊢ v1 : τ1 Γ, x : Gen(τ1,Γ) ⊢ e2 : τ2
Γ ⊢ let x = v1 in e2 : τ2

∀τ1.(σ1 > τ1 → Γ ⊢ v1 : τ1) Γ, x : σ1 ⊢ e2 : τ2
Γ ⊢ let x = v1 in e2 x : τ2

図 8. Let項の型付け規則

5.2 let多相
多相の型を使って let多相を定式化する。let多相を加えた単純型付き λ計算の型と型スキーム、

そして構文を図 7に示す。図 7において、Let項の定義は let x = v1 in e2 xのように e2が xを受け
取るような形になっているが、これは e2の中に変数 xが表れていることを示すものである。本稿
では、副作用は扱っていないが、将来的に限定継続命令を導入することを視野に入れて、let 文は
value restriction を採用している。
以下に、値と項のAgdaによる定義を示す。

mutual
data value[_]_ {tvar : Set} (var : ts[ tvar ] → Set) : typ[ tvar ] → Set where

Var : {σ1 : ts[ tvar ]} → (x : var σ1) → {τ1 : typ[ tvar ]} → �� 変数
(p : inst σ1 τ1) → value[ var ] τ1

Num : (n : N) → value[ var ] Nat �� 自然数
Fun : {τ1 τ2 : typ[ tvar ]} → (e1 : var (Typ τ2) → term[ var ] τ1) → �� ラムダ抽象

value[ var ] (τ2 ⇒ τ1)

data term[_]_ {tvar : Set} (var : ts[ tvar ] → Set) : typ[ tvar ] → Set where
Val : {τ1 : typ[ tvar ]} → value[ var ] τ1 → term[ var ] τ1 �� 値
App : {τ1 τ2 : typ[ tvar ]} → �� 関数適用

(e1 : term[ var ] (τ2 ⇒ τ1)) → (e2 : term[ var ] τ2) → term[ var ] τ1
Let : (σ1 : ts[ tvar ]) → {τ2 : typ[ tvar ]} → �� 多相の let

(v1 : {τ1 : typ[ tvar ]} → inst σ1 τ1 → value[ var ] τ1) →
(e2 : var σ1 → term[ var ] τ2) → term[ var ] τ2

2節における値と項の定義では、var は型を受け取る関数であったが、多相の型を扱うため var は
型スキームを受け取るようになっている。Varと Letの定義以外は、2節の定義を型スキームを扱う
ように書き換えただけである。Varと Letにおいて、多相の型を扱うための新しい関係 instという
ものが表れている。inst σ τ は、多相の型 σが単相の型 τ に具体化することができることを表す関
係である。この関係を定義するため、型の代入も定義しているが、項の代入の定義とほとんど同じ
方法で行なっているため、説明は省略する。これを踏まえて定義を見ていこう。Varは var σ1型の
変数 xを受け取ったら、σ1型が τ1型に具体化できるとき τ1型の変数を表す。Letでは、まず型 σ1

を受け取る。これは、Letによって定義される変数の多相の型を表している。次に v1を受け取るが、
v1 は多相の型 σ1 を持つ値である。v1 が確かに多相の型を持っていることを保証するため、v1は
σ1と inst関係にある任意の τ1型を持つようにしている。続いて e2を受け取る。e2は var σ1型の変
数を受け取ったら τ2型の項を返す関数である。それらを受け取ったら Letの項を表すという定義で
ある。この定義は let x = v1 in e2 xを意味している。e2の項を Agda上の関数として扱うことで、
e2の中の変数 xの束縛を表現することができているとわかる。
本稿の定義における Let項の型付け規則は、一般的な型付け規則とは少々異なる形をしている。

図 8は、左が一般的な Let項の型付け規則で、右が本稿における Let項の型付け規則である。ここ

で、Gen(τ1, Γ)は、τ1に自由に表れていて Γに含まれない型変数を束縛して多相の型として扱う関
数を、σ1 > τ1は inst σ1 τ1を表す。一般的な Let項の型付け規則では、関数Genというものを定義
して変数の束縛を関数で行うが、本稿ではHOASを採用しており、全ての変数の束縛はメタ言語の
レベルで行われる。そこで、本稿では多相の型から始めて、それを具体化したどの型も持つ、とい
う逆のアプローチで型付け規則を定義した。

5.3 CPS変換後の項
let多相の型付け規則を前章のように行なったことで、CPS変換の前後で項の定義を別に行う必

要が出てきた。その理由として、まず Let項の CPS変換の定義を見ていく。

[[let x = v in e x]] = λκ. let x = [[v]]v in [[e x]] @κ

[[let x = v in e x]]′ = λk. let x = [[v]]v in [[e x]]′@ k

図 9. let多相の CPS変換

図 9からわかるように、let x = v in e xの CPS変換は vと e xそれぞれについて再帰して CPS

変換を行なっていく。ここで vは図 8の型付け規則より（CPS変換前の）inst関係を受け取らない
限り値にはならないことに注意しよう。今、図 9の右辺はCPS変換後の項であるため、この Let項
で得られる inst関係は CPS変換後のものである。vを値にするためには、この CPS変換後の inst

関係を何らかの方法でCPS変換前の inst関係に戻す必要がある。これができて初めてCPS変換が
可能となる。しかし、3節のように型を変換前後で異なる形に変換してしまうと、CPS変換前後で
inst関係の定義も変わることになり、CPS変換後から前への inst関係の変換が行うことが困難であ
ることがわかった。
そこで、この節ではCPS変換の前後で項の定義を変更し、CPS変換後の構文をCPS変換前の構

文と一対一に対応させ、両者で同じ型を使うことで inst関係の変換を避けることにした。CPS変換
後の項の構文は図 10のようになる [7]。

継続 k ::= x | λx. e
値 v ::= c | x | λxk. e
項 e ::= v | e1 e2 k | k v | let x = v in e x

図 10. CPS項

Agda では以下のようになる。cpsvalue[var ]τ は τ 型の値を、cpsterm[var ]Natは継続を適用する
ため最終的に Nat型となる項を、そして cpscont[var ]τ ⇒ Natは τ 型の項を受け取ったら Natを返
すような継続を示す。継続はCPS変換を行うことで生み出される項であるため、項や値からは区別
して定義する。ラムダ抽象 Funは CPS変換を行うと、引数と継続を受け取って、最終的には継続
を適用して返す、という形に変換されるため、e1はこれまでの引数に加えて、必ず継続を表す var

(Typ (τ1 ⇒ Nat))型の変数も受け取るような Agdaの関数となっている。関数適用 Appは CPS変
換を行うと、関数と引数に加えて継続 cも受け取るようになっている。Let項についてはCPS変換
前の定義と同じであるが、最終的に Natが返るような項となっていることがわかる。このように、
項の定義そのものに CPS変換が反映されているため、CPS変換前の項と同じ型を扱うことができ
るようになり、Let項の CPS変換を定式化することができるようになった。以後この項を CPS項
と呼び、対応して CPS変換前の項をDS (Direct Style)項と呼ぶ。

mutual
data cpscont[_]_⇒Nat {tvar : Set} (var : ts[ tvar ] → Set) : typ[ tvar ] → Set where

Var : {τ1 : typ[ tvar ]} → (k : var (Typ (τ1 ⇒ Nat))) → �� 継続を表す変数
cpscont[ var ] τ1 ⇒Nat

Fun : {τ1 : typ[ tvar ]} → (e1 : var (Typ τ1) → cpsterm[ var ]Nat) → �� 継続を表すラムダ抽象
cpscont[ var ] τ1 ⇒Nat

data cpsvalue[_]_ {tvar : Set} (var : ts[ tvar ] → Set) : typ[ tvar ] → Set where
Var : {σ1 : ts[ tvar ]} → (x : var σ1) → {τ1 : typ[ tvar ]} → inst σ1 τ1 → �� 変数

cpsvalue[ var ] τ1
Num : (n : N) → cpsvalue[ var ] Nat �� 自然数
Fun : {τ1 τ2 : typ[ tvar ]} → �� 継続までを含むラムダ抽象

(e1 : var (Typ τ2) → var (Typ (τ1 ⇒ Nat)) → cpsterm[ var ]Nat) →
cpsvalue[ var ] (τ2 ⇒ τ1)

data cpsterm[_]Nat {tvar : Set} (var : ts[ tvar ] → Set) : Set where
App : {τ1 τ2 : typ[ tvar ]} → �� 継続の適用までを含む関数適用

(v1 : cpsvalue[ var ] (τ2 ⇒ τ1)) → (v2 : cpsvalue[ var ] τ2) →
(c : cpscont[ var ] τ1 ⇒Nat) → cpsterm[ var ]Nat

Ret : {τ1 : typ[ tvar ]} → �� 継続の適用
(c : cpscont[ var ] τ1 ⇒Nat) → (v : cpsvalue[ var ] τ1) → cpsterm[ var ]Nat

Let : (σ1 : ts[ tvar ]) → �� let多相
(v1 : {τ1 : typ[ tvar ]} → inst σ1 τ1 → cpsvalue[ var ] τ1) →
(e2 : var σ1 → cpsterm[ var ]Nat) → cpsterm[ var ]Nat

5.4 代入規則と簡約規則
代入規則と簡約規則について定める。DS項とCPS項のそれぞれについて代入規則、簡約規則の

定義が必要となる。
まず、DS項についての規則について説明する。代入規則は、型が型スキームに変化するのみで 2

節の定義と基本的には同じである。Let項の代入規則は、図 11のようになる。

(λy. v1 y)[v] 7→ v′1 ∀x((λy. e2 y x)[v] 7→ e′2 x)

(λy. (let x = v1 y in e2 y x))[v] 7→ let x = v′1 in e′2 x

図 11. Let項の代入規則

Let項の定義において e2はAgdaの関数の形をしているため、項の形にするには何らかの引数を
与える必要がある。そのため、代入の再帰部分において任意の xを引数として与える形で定義して
いる。
次にDS項の簡約規則について説明する。簡約規則についても、型が型スキームに変化するのみ

で 2節の定義と基本的には同じであるが、Let項の簡約を追加する必要がある。Let項の簡約規則は、
図 12のようになる。

(λx. e x)[v] 7→ e′

let x = v in e x → e′

図 12. Let項の簡約規則

let x = v in e xは、eに含まれる xに vを代入して e′が得られるならば、e′に簡約される。図 12

の規則は１ステップ簡約であり、０ステップ以上の簡約は 2節と同様に定義する。
次に、CPS項の代入規則について説明する。CPS項には、通常の関数適用と継続の適用の２種

類の関数適用がある。このうち、前者は引数に加えて必ず継続も渡す形になっている。それに合わ
せて、代入規則も引数と継続両方を代入する規則を定義する。継続も含む代入を以下のように表記
する。

e[v, c] 7→ e′

これは「e の中の引数部分（最初の引数）に v を代入し、継続部分（ふたつ目の引数）に c を代
入したら e′ となる」と読む。この代入規則の定義は、代入されるものが二つになるのみで、考え方
は 2節の代入規則と変わらないため、説明は省略する。また、一つの代入のみを扱う代入規則も、
継続を扱う際には必要となったため、定義を行なった。これも、2節と同じ考え方であるため、説
明は省略する。
CPS項の簡約規則を図 13に示す。簡約規則は、継続付きの β簡約と継続の適用の２種類からな

る。CPS項の簡約規則にはフレームの規則はない。これは、CPS 変換後の項は全てが末尾呼び出
しになっており、常に簡約子が一番、外側に現れているからである。

(λx k. e)[v, c] 7→ e′

((λx k. e) v c) → e′
(λx. e)[v] 7→ e′

(λx. e) v → e′

図 13. CPS項の β 簡約規則と継続適用規則

図 13は１ステップの簡約を表す定義であり、０ステップ以上の簡約規則を定義する必要がある。
単純型付き λ計算において、CPS変換の正当性の証明は１方向のみの複数簡約のみで終了すること
ができるというものであったが、let多相を加えたところ、より一般的な関係を作らざるを得なく
なった。後の、CPS変換の正当性の証明について説明する際に、両方向の簡約をしなければ証明が
できないような状況が出現する。したがって今節では、CPS項については０ステップ以上簡約はど
ちらの方向の簡約であっても繋がっていれば同値であるという定義になっている。
以後、DS項の１ステップ簡約を→、CPS項の０ステップ以上簡約の同値関係を∼で表す。

5.5 let多相を含む単純型付き λ計算のCPS変換の正当性の証明
let多相で拡張した単純型付き λ計算の CPSの正当性を表す定理は以下のようになる。

定理 3 (CPS変換の正当性) 任意のDS項 e, e′について e → e′ が成り立つなら、任意の schematic

な継続 κについて [[e]] @κ ∼ [[e′]] @κ が成り立つ。

ここで、4節と同様に schematicな継続というものが出てくるが、定義は 4節のものとは少々異
なるものになっており、定義は以下の通りである。

(λy. v1)[v] 7→ v′1

(λy. κ@ v1 y)[v] 7→ κ@ v′1

ある値 v1が vの代入により v′1になるならば、代入前後の値にそれぞれ継続 κを適用しても代入
関係が維持される、という性質を表している。
この schematicの定義で、Let項以外の項についての正当性の証明を行うことができている。単相

の型による単純型付き λ計算における schematicもこちらの定義にできると考えているが、まだ実
際に確認は行っていない。
定理の証明について説明する。4節の証明に、Let項の証明を付け加えることになるが、cpsEtaEta′

における Letの場合以外は 4節と同様に証明が終了する。その際、4節における補題 eSubstや kSubst

を１つにまとめたような補題を証明する必要がある。これは、CPS項の定義により、代入規則を継
続の代入も同時に行うように定義したものが存在するため、これに沿った補題が必要となるからで
ある。具体的な証明は、4節と同様である。2節と同じような、一つの値の代入に対する補題も、4

節と同様に示す。次に Let項の簡約の場合については、上の補題と一回の簡約規則を適用すること
で示すことができた。

次に、cpsEtaEta′の Letの場合の証明について説明する。ここで、示さなければならないものは
以下である。

let x = [[v1]]v in [[e2 x]]
′@(λv. κ v) ∼ let x = [[v1]]v in [[e2 x]] @κ

両辺 Let項であるため（手動の証明では）代入によって簡約することができる。このとき、右辺
の簡約は 4節における Reduce*に含まれる簡約とは逆向きの簡約であるため、∼は双方向の簡約に
ついての同値関係として定義されている。代入による簡約を図 14に示す。

左辺の簡約 (λx. ([[e2 x]]
′@(λv. κ v)))[[[v1]]v] 7→ [[e′2]]

′@(λv. κ v)

右辺の簡約 (λx. ([[e2 x]] @κ))[[[v1]]v] 7→ [[e′2]] @κ

図 14. Let項の簡約

ここで、両辺の簡約結果にある e′2 は、λx. (e2 x)[v1] 7→ e′2 が成り立つような項であり、すなわ
ち e2 に含まれる xに v1 を代入した結果の項を表す。両辺の簡約を行なったことで示すべき式は
[[e′2]]

′@(λv. κ v) ∼ [[e′2]] @κとなる。これは、cpsEtaEta′の再帰によって示すことができる。
しかし、Agdaによる定式化では、e′2を具体的に与えることができない。代入される項を場合分

けすることができれば、それぞれの場合で具体的に e′2を考えることができるはずだが、代入される
項はAgdaの λ式となっている。なぜなら、この定式化は PHOASによって行われているからであ
る。したがって、Agdaにおいて代入される項、λx. (e2 x)の場合分けを行うことができないため、
この項に依存した項である e′2を具体的なコードとして書くことができないのである。何らかの具体
的な引数を与えてやれば λ式の本体を取り出し場合分けを行うことができるが、今回与えるべき引
数の型は、var σ(σは適当な型スキーム)のようなもので、具体的に作ることは難しい。このため、
定理 3を示すことができなくなってしまった。
図 14は、単に let文の progress の性質を述べているにすぎない。しかし、PHOAS を使っている

ため、それを示すことができず、行き詰っている格好である。これは、PHOAS を使うと多くの証
明が簡潔にできている一方で、依然として困難も残っていることを示している。
今のところ、progress を示す方法が見つからないため、本稿ではそれを仮定する形で定式化した。

仮定 1 (Let項の同値性) 変数を受け取ったらCPS項を返すような任意の関数e1, e2について∀x(e1 x ∼
e2 x) が成り立つなら、任意の CPS項の値 vについて let x = v in e1 x ∼ let x = v in e2 x が成り
立つ。

まず、任意の xのもとで、e2 xについて cpsEtaEta′ の再帰を行う。すると、任意の xについて
[[e2 x]]

′@(λv. κ v) ∼ [[e2 x]] @κが得られる。この式に対して仮定 1を適用すると、示したい式が得
られる。以上より、仮定 1を使えば、e′2を具体的に与えることなく、証明を終了することができる。
仮定 1の証明は、今のところできていない。証明ができない理由は、∀x(e1 x ∼ e2 x)がAgdaにお
いて λ式として表されており、場合分けができないからである。なぜ λ式を作らなければならない
かというと、e1, e2は関数であるため、何らかの引数を与えてやらなければCPS項にはならず、∼
関係を仮定することができないからである。今後、さらに検討するが、これは PHOAS の本質的な
難しさにつながっていると思われる。

6 関連研究
CPS 変換の正当性を定式化した研究に、[12]がある。これは定理証明支援系言語 Isabelle/HOL

によって FOASを用いて証明を行っているため、α変換を全て書き下している。また、[10]では、

対象言語に現れる λ抽象のための de Bruijn index と、CPS変換時に現れる継続のための de Bruijn

index とを分けて取り扱うことで α変換での問題を回避しつつ証明を行っている。
PHOASに似た変数束縛の表現方法にweak HOASがある。[14]では、weak HOASを用いてオブ

ジェクト指向言語である Featherweight Java の定式化を Coq で行っている。PHOASそのものの
応用例に [15]がある。これはグラフの閉路などを取り扱う際、ミュータブルな辺や頂点のリストを
使用する従来のアプローチとは異なり PHOASを用いて表現するものである。

7 まとめと今後の課題
本稿では、単純型付き λ計算におけるDanvy/Filinskiの one-passのCPS変換を定式化し、その

正当性を Agdaで証明した。また、それを let多相の入った体系においても成り立つようにするた
め、CPS項を定義し同様に証明した。いずれの体系においても定式化の方法に PHOASを採用し、
変数束縛はメタ言語の Agdaに処理させた。これにより、変数束縛による証明の複雑化を避け、手
で書いた証明とほぼ対応した定式化をすることができた。
しかし、let多相の入った体系では高階関数の中身に立ち入ることができないという問題に直面

し、明らかに成り立つと思われる仮定を置く必要があった。この仮定を外せるかどうかの検討は今
後の課題だが、PHOASの本質的な難しさに起因しているのではないかと予想している。
今回、単相の部分については十分に簡潔にCPS変換の証明を行えることがわかったので、今後、

shift/resetを加え、selective CPS変換 [3] の定式化につなげたいと考えている。

